スターコンプリメントテクニックと最小固有値が-2以上のグラフの生成

2012/1/26

概 要

最小固有値が -2 以上のグラフの生成には、ルート系の部分集合（ノルム 2 のベクトルからなるユークリッド空間の部分集合）が用いられる。その様なグラフの隣接行列はそれらベクトルの部分集合のなす行列 D によるグラム行列で表される。グラフの最小固有値研究の場面では、こようなグラフを生成する必要に（頻繁に）駆られる。D. Cvetković, P. Rowlinson, S.K. Simić 氏達によって、スターコンプリメントテクニックを用いた極大例外型グラフの分類が成された (cf. [1])。このテクニックを紹介する。

[1] Star complements and exceptional graphs. Linear Algebra and its Applications (2007), 423(1), 146–154.

1 Star complemet technique

ある特定の固有値を持つグラフが（重複度も含め）自由に生成できたら嬉しい場面が、あるのではないでしょうか。ここではその様な場面で有用な **star complement tecnique** を紹介します。

Definition: Let Γ be a graph with $V(\Gamma) = \{1, \ldots, n\}$. Let P be the orthognal projection of $\mathbb{R}^n$ onto $\mathscr{E}(\mu)$, where $\mathscr{E}(\mu)$ is the eigenspace of $A(\Gamma)$ for the eigenvalue μ of $A(\Gamma)$. Then a subset X of $V(\Gamma)$ satisfying the following condition is called a **star set** for μ of Γ:
the vectors $P\mathbf{e}_j$ $(j \in X)$ form a basis for $\mathscr{E}(\mu)$, where $\{\mathbf{e}_1, \ldots, \mathbf{e}_n\}$ is the standard basis of $\mathbb{R}^n$.

Definition: Let Γ be a graph with $V(\Gamma) = \{1, \dots, n\}$ and an eigenvalue μ. Let X be a star set for μ of Γ. Then the subgraph $\Gamma - X$ of Γ is called the **star complement** for μ corresponding to X.

Let Γ be a graph with adjacency matrix $\begin{pmatrix} A_X & B^T \\ B & C \end{pmatrix}$, where X is a star set for an eigenvalue μ of Γ. Then we define a bilinear form on $\mathbb{R}^{n-|X|}$ by

$$\langle \mathbf{x}, \mathbf{y} \rangle_X = \mathbf{x}^T (\mu I - C)^{-1} \mathbf{y},$$

and denote the columns of B by $\mathbf{b}_v$ $(v \in X)$.

Theorem 1: Suppose that μ is not an eigenvalue of the graph Γ'. Then there exists a graph Γ with a star set X for μ such that $\Gamma - X = \Gamma'$ if and only if the characteristic vectors $\mathbf{b}_v$ $(v \in X)$ satisfy

(i) $\langle \mathbf{b}_v, \mathbf{b}_v \rangle_X = \mu$ for all $v \in X$,

(ii) $\langle \mathbf{b}_u, \mathbf{b}_v \rangle_X \in \{-1, 0\}$ for all pairs u, v of distinct vertices in X.

2 Star complemet technique の流れ

例えば、λ という固有値を持つグラフを作りたいとする。このとき、次の流れで、固有値 λ（重複度 m）を持つグラフをつくる：

(i) Let Γ be a graph without eigenvalue λ.

(ii) Compute $S := \{\mathbf{b} \in \{0,1\}^{|V(\Gamma)|} \mid \mathbf{b}^T (\lambda I - A(\Gamma))^{-1} \mathbf{b} = \lambda\}$.
※ ここで $S = \emptyset$ となれば、Γ に頂点を加えることで、固有値 λ をもつグラフは生成出来ない事を意味する。

(iii) Let T be a subset of S such that $\mathbf{u}^T (\lambda I - A(\Gamma))^{-1} \mathbf{v} \in \{0, -1\}$ for all pairs $\mathbf{u}, \mathbf{v}$ of distinct vectors in T.

(iv) Let G be a graph with adjacency matrix

$$\begin{pmatrix} \lambda I - B^T (\lambda I - A(\Gamma))^{-1} B & B^T \\ B & A(\Gamma) \end{pmatrix}.$$

※ Theorem 1 から、G が固有値 λ（重複度 m）を持つことがわかる。

以下は magma による、固有値 $-1-\sqrt{2}$（重複度 2）を持つグラフを生成するプログラムである。

MAGMA:

```
//準備
n:=7;
Q:=Rationals(); P<x>:=PolynomialRing(Q);
//グラフの定義
G:=Graph<{@i: i in [1..7]@}|
    {@ {1, 5}, {1, 6}, {2, 3}, {2, 5}, {2, 7}, {3, 4}, {3, 5},
    {3, 6}, {3, 7}, {4, 5}, {4, 6}, {5, 6}, {6, 7} @}
    >;
//G:=PathGraph(n);
Cp:=CartesianPower([0,1],n);
Vecs:=[Matrix(n,[cc : cc in c]) : c in Cp];
//隣接行列
M:=AdjacencyMatrix(G);
//Zero に-1-√2 を持つ多項式
p:=x^2+2*x-1;
f:=P!CharacteristicPolynomial(M);

F:=SplittingField(p);
C:=ChangeRing(M,F);
B:=Basis(F);
_:=exists(r){b:b in B| b notin Q};

MR:=MatrixRing(F,Order(G));
W:=(r*MR!1-M)^-1;
V:=[ChangeRing(v,F) : v in Vecs];
X:=[v :v in V |
    (v*W*Transpose(v))[1,1] eq r
];

S2:=[Setseq(s):s in Subsets(Seqset(X),2)];
Ts:=[s: s in S2 |
    (s[1]*W*Transpose(s[2]))[1,1] in [0,-1]
];

T:=Random(Ts);
v1:=T[1];
v2:=T[2];

Bt:=VerticalJoin(ChangeRing(v1,F),ChangeRing(v2,F));
L2:=VerticalJoin(Bt,C);
```

```
B:=Transpose(Bt);
K:=-Bt*W*B;
for i in [1..2] do
        K[i,i]:=0;
end for;
Ut:=VerticalJoin(K,ChangeRing(B,F));
L2:=HorizontalJoin(Ut,L2);
f2:=P!CharacteristicPolynomial(L2);
f2/p in P;
Factorization(f2);


/*
> Spectrum(Graph<{@i:i in [1..9]@}|L2>);
[ <-2.41421356237309504880168872420969807856967187537694807317638, 2>,
<0.41421356237309504880168872420969807856967187537694807317638, 2>,
<0.16142518519415694875674044541390708865916260872093420786805, 1>,
<1.43108998400099700379096592141838254487761558765778855602668, 1>,
<-0.9999999999999999999999999999999999999999999999999999998961, 1>,
<-1.6962904470265927588328135139618418970024894842153498646592, 1>,
<5.10377527783143880628510714712955226346571128783662710072160, 1> ]
*/
```